```
// Dialogic-code for Railways Journey Planner (this English version code is not actually used,
// but translated from Dutch just in order to make it understandable for the non-Dutch reader)

version 0.95
language nl-NL    // default language
gendiagrams false // debug: regex diagrams, just for checking
//permissions ['NAME', 'DEVICE_PRECISE_LOCATION']
permission_required false

// Multi-language strings (actually not used in this version)
$confirm =
  nl-NL: "Ik verstond {0}"
  en-GB: "I heard {0}"
$askArrivalStation =
  nl-NL: "Waar wilt u heen? Noem een station|Noem aankomststation"
  en-GB: "To which railway station would you like to travel?"
  de-DE: "Zum welchen Bahnhof möchten Sie reisen?"
$geenStation =
  nl-NL: "Ik verstond {0}. Ik herken dit niet als stationsnaam"
  en-GB: "I heard {0}, but do not recognize this as a railway station name"
$explanationAboutStations =
  nl-NL: "Geef de naam van een station op"
  en-GB: "Enter the name of the railway station"
$bevestigAankomstStation =
  nl-NL: "U wilt naar {0}"
  en-GB: "You want to go to {0}"
$ConfirmStationSelection =
  nl-NL: "U wilt van {0} naar {1}"
  en-GB: "You will travel from {0} to {1}"
$askDepartureStation =
  nl-NL: "Vanaf welk station vertrekt u naar {0}?|Noem vertrekstation"
  en-GB: "From which railway station do you go to {0}|?"
$bevestigVertrekStation =
  nl-NL: "U vertrekt vanaf {0}"
  en-GB: "You will depart from {0}"

synonyms
  (a an)
  (leave depart)


dialog Reisplanner
```

```
(
  out string [] Dstat, // departure station, 2 entries: full name + technical abbreviation
  out string [] Astat,
  out DateTime TravelDate,
  out DateTime TravelMoment,
  out bool D_or_A,
  out string TravelMomentDA,
  out bool FreeSubscription,
  out bool InclReservationRequired,
  out bool AdditionalOptions,
  out bool NeedExtraTransferTime,
  out int ExtraTransferMinutes
)

customcode CustomCode

signatures // methods implemented in custom code. Templates are generated automatically
  string [] CheckStation (string [] station)
  void DeterminePossibleJourneys (string dStat, string aStat, string via, Date journeyMoment, bool d_or_a,
    int xtraTransferMinutes, bool inclResrvationRequired, bool freeSubscription, bool transferOK,
    out int nrOfAlternatives, out int curAlternative, out string [,] prices)
  void ObtainJourneyDetails (int index, out string dTime, out string aTime, out string platformNo,
    out int nrOfTrainChanges, out string interchangeStations, out string journeyTime,
    out string journeySchedHtml, out string journeySchedText, out string interchangeHtml, out string
interchangeText)

mongo_url_loc
  "mongodb://localhost/reisplanner"
mongo_url_dev
  "mongodb+srv://${mongoUsr}:${mongoPwd}@dereisplanner-jmlem.gcp.mongodb.net/test"
mongo_url_pro
  `mongodb://${mongoUsr}:${mongoPwd}@dereisplanner-shard-00-00-jmlem.gcp.mongodb.net:27017,
   dereisplanner-shard-00-01-jmlem.gcp.mongodb.net:27017,
   dereisplanner-shard-00-02-jmlem.gcp.mongodb.net:27017/
test?ssl=true&replicaSet=DePeisplanner-shard-0&authSource=admin&retryWrites=true`

activities travel_advice, additional_options

{
  // global variables
  Date travelDateTime = <today>;
  string timeString = "";
```

```
85      string dateString = "";
86      string dayName;
87      int hhh = -1;
88      int mmm = -1;
89      int dayPart;
90      bool started = false;
91      bool reported = false;
92      bool user_done = false;
93      bool get_journey_plan = false;
94      bool do_check_early = false;
95      string journeySchedHtml = ""; // browser html
96      string journeySchedText = ""; // smart device screen text
97      string interchangeHtml = ""; // browser html
98      string interchangeText = ""; // smart device screen text
99      string journeyTime = "";
100     int nrOfTrainChanges = 0;
101     string interchangeStations = null;
102     int nrOfAlternatives = 0;  // number of journey alternatives found
103     int curAlternative = -1;   // index of selected journey
104     int optRmhNo;        // index of optimal journey
105     string [,] prices = null;
106     string copyright = 'U+00A9';
107     //string backArrow = '<<';
108     //string forwArrow = '>>;
109     //string pricesIco = 'U+20ACU+20AC' // double euro sign
110     Date timetableExpiryDate = MakeDate (2019, 12, 8);
111
112      // Triggers -- a trigger is a piece of code that is executed by the Dialog Driver
113      //             each time its precondition is (reset from false to) true
114
115     trigger Initialize () precondition (!started)
116     {
117       [!nl-NL!] // set language to Dutch (sort of obsolete now, dynamic switching is not possible. For browser only)
118       started = true;
119       enable activity travel_advice exclusive ; // multiple activities can be active at the same time
120       say ('Hello this is your journey planner|Journey Planner') // => converted to SimpleResponse speech|text
121     }
122
123
124     trigger Finalize () precondition (user_done)
125     activity travel_advice
126     {
```

```
    card ("Possible journey", null, journeySchedText, null, null); // BasicCard
    halt("Thank you, I wish you a pleasant journey");
}


// Questions -- a question is executed when both its argument is
//              'unknown' and the possible precondition is true. It contains patterns + code

question AskArrivalStation (Astat) before AskDepartureStation // 'Astat' is only asked for when its
// value (and that of Question parameters in general) is unknown. Such values can be in the following
// states: unknown (unk), set, i.e. they have a value (set), or irrelevant (irv). These states can be
// explicitly assigned. Only 'unk' values will be asked for, and then their state will change to 'set'.
activity travel_advice
ask $askArrivalStation
explain $explanationAboutStations // explanation text is spoken after wrong answer only
{
  [ DateTime date = <today> ] // script code is in [ square brackets ]
  ( ( i want to ( travel | go ) ) to !! // !! acts like a Prolog cut, preventing backtracking
    StationName (out Astat) // StationName is a Pattern, see below.
  | ( i ( want | have to ) travel
      ( on? Date(out date, out dateString, out dayName)
        [ TravelDate = date ] // variable value settings are undone upon failure / backtracking
      )?
      ( Moment (date == <today>, out TravelMoment, out timeString) // <today> etc. are predefined values
      )?
      ( from StationName (out Dstat) )?
    )
    ( ( ( travel | leave | go )? to )
      StationName (out Astat)
      [ if (is_set (TravelMoment)) D_or_A = true ] // assumption: time supplied is departure time
    | arrive ( at station? |in )
      StationName (out Astat)
    )
  )
  [ if (is_unk (TravelDate) && is_set (TravelMoment)) // assumption: today
      TravelDate = <today>
  ]
}


question AskDepartureStation (Dstat)
activity travel_advice
```

```
ask ($askDepartureStation, Astat [0])
//explain $explanationAboutStations
{
  ( [?nl-NL?]
    ((ik wil?)? ( vertrekken | vertrek))? vanaf
  | [?en-GB?]
    ((i ( want to)?)? ( leave | depart))? from
  )?
  StationName (out Dstat)
}


trigger ConfirmStationSelection () precondition (any_set (Dstat, Astat))
{
  if (all_set (Dstat, Astat))
  {
    if (Dstat [1] == Astat [1])
    {
      say ("The departure station is equal to the arrival station");
      say ("Please provide both station names again");
      set_unk (Dstat, Astat) // => set to unknown, so they will be asked for again
    }
    //else
    //  say ($ConfirmStationSelection, Dstat [0], Astat [0])
  }
  else if (is_set (Dstat))
    say ($confirm, Dstat [0])
  else if (is_set (Astat))
    say ($confirm, Astat [0])
}


question AskWhen (TravelDate) after AskDepartureStation
activity travel_advice
ask ("When would you like to travel?|")
suggest('now', 'tomorrow', 'day after tomorrow')   // @N => N milliseconds <break> (seconds if N < 10)
explain "Say: now @600 tomorrow @600 friday @600 or next monay @800 of provide a date"
{
  [ TravelDate = <today> ]  // assumption
  ( Moment (true, out TravelMoment, out timeString)
    [ D_or_A = true ]
  | ( (i want to travel)? on | (ik wil reizen)? op )?
```

```
211         Date (out TravelDate, out dateString, out dayName)
212         ( Moment (TravelDate == <today>, out TravelMoment, out timeString)
213           [ D_or_A = true ]
214         )?
215       )
216     }
217
218
219     trigger CheckDateMoment0 () before CheckDateMoment1 precondition (all_set (TravelDate, TravelMoment, D_or_A))
220     activity travel_advice
221     {
222       if (hhh <= 6) do_check_early = true;
223     }
224
225
226     trigger CheckDateMoment1 () precondition (!do_check_early && all_set (TravelDate, TravelMoment, D_or_A))
227     activity travel_advice
228     {
229       set_unk (TravelMomentDA);
230
231       if (TravelDate > timetableExpiryDate)
232       {
233         say (`The current timetable expires on {0}'|Timetable expires on {0}`, timetableExpiryDate);
234         // text before | is spoken, after | on display. If | at end, display text = speech text
235         set_unk (TravelDate, TravelMoment);
236       }
237       else
238       {
239         travelDateTime = MakeDateTime (TravelDate, TravelMoment);
240         timeString = TimeToString (travelDateTime);
241
242         if (TravelDate == <today>)
243           dateString = "today";
244         else
245           dateString = Format ("on {0} {1}", DayName (TravelDate), DateToString (TravelDate));
246
247         TravelMomentDA = Format ("{0} at {1}", (D_or_A ? "Departure" : "Arrival"), timeString);
248       }
249     }
250
251
252     question ConfirmMomentPriorTo6am () precondition (do_check_early)
```

```
253     activity travel_advice
254     ask (`Do you mean {0} in the morning of {0} in the afternoon? Say morning or afternoon|
255          {0} in the morning of {0} in the afternoon?`, timeString)
256     suggest('morning', 'afternoon')
257     {
258       <any>* // zero or more arbitrary words may precede
259       ( ( morning | ochtend | s? ochtends )
260         [ say ("Ok, in the morning") ]
261       | ( afternoon | middag | s? smiddags )
262         [ TravelMoment = MakeDateTime (TravelDate, hhh+12, mmm);
263           timeString = TimeToString (TravelMoment);
264           say ("Ok, in the afternoon")
265         ]
266       )
267       [ do_check_early = false ]
268     }
269
270
271     question AskIfMomentIsDepartureOrArrivalTime (D_or_A) after AskWhen
272     activity travel_advice
273     ask ("{0}", is_set (TravelMoment)
274       ? Format ("Is {0} a departure time or an arrival time?|Departure time or an arrival time?", TimeToString
275   (TravelMoment))
276       : "Do you want to provide a departure time or an arrival time?|Departure time or arrival time?")
277     suggest('Departure time', 'Arrival time')
278     {
279       <any>*
280       ( departure time? [D_or_A = true]
281       | arrival time? [D_or_A = false]
282       )
283     }
284
285     question AskWhatTime (TravelMoment) after AskIfMomentIsDepartureOrArrivalTime
286     activity travel_advice
287     ask ("What time do you want to {0}?", D_or_A ? "leave" : "arrive")
288     suggest((D_or_A && TravelDate == <today>) ? ['now', 'within half an hour', 'within an hour'] : [])
289     {
290       Moment (TravelDate == <today>, out TravelMoment, out timeString)
291     }
292
293
294     question AskAdditionalOptions (AdditionalOptions) precondition (all_set (Dstat, Astat, TravelDate,
```

```
295   TravelMoment))
296     activity travel_advice
297     ask ("Do you want to specify additional options?|")
298     suggest('Yes', 'No')
299     { YesNo (out AdditionalOptions)
300       [ if (AdditionalOptions) // only select questions etc. beloning to activity 'additional_options'
301           enable activity additional_options ;
302         else // irrelevant => will not be asked:
303           set_irv (FreeSubscription, InclReservationRequired, NeedExtraTransferTime, ExtraTransferMinutes)
304       ]
305     }
306
307     question AskExtraTransferMinutesRequired (NeedExtraTransferTime) after AskAdditionalOptions
308     activity additional_options
309     ask "Do you need extra time for changing trains?|Extra time for changing trains?"
310     suggest('Yes', 'No')
311     {
312       YesNo (out NeedExtraTransferTime)
313       [if (!NeedExtraTransferTime) set_irv (ExtraTransferMinutes)] // irrelevant => will not be asked
314     }
315
316
317     question AskExtraTransferMinutes (ExtraTransferMinutes) after AskAdditionalOptions
318     precondition (NeedExtraTransferTime)
319     activity additional_options
320     ask "How many extra minutes dou you need?"
321     suggest(5, 10, 15, 20)
322     {
323       @ExtraTransferMinutes:<int>    // <int> matches any integer, <int:n..m> between n and m inclusive
324       [ if (ExtraTransferMinutes > 20)
325           fail("20 minutes extra is maximum|20 is maximum")
326       ]
327     }
328
329     // The notation @varname:pattern means that variable 'varname' will be set to the value of 'pattern'.
330     // Example: @s:(john mary+ fred) will be set to "john mary mary fred" if that was what the user said.
331
332     question AskFreeSubscription (FreeSubscription) after AskExtraTransferMinutesRequired
333     activity additional_options
334     ask "Do you have a Altijd Vrij of Trein Vrij subscription?|"
335     suggest('Yes', 'No')
336     {
```

```
337       YesNo (out FreeSubscription)
338     }
339
340
341     question AskReservationRequired (InclReservationRequired) after AskFreeSubscription
342     activity additional_options
343     ask `Do you want trains for which reservation is required to be taken into account?|
344         Take into account trains for which reservation is required?`
345     suggest('Yes', 'No')
346     {
347       YesNo (out InclReservationRequired)
348     }
349
350
351     trigger SpecsCompleet1 () precondition (!any_unk(*)) // no more unknowns, enough questions answered
352     activity travel_advice
353     {
354       DeterminePossibleJourneys (Dstat [1], Astat [1], null, travelDateTime, D_or_A, ExtraTransferMinutes,
355         InclReservationRequired, FreeSubscription, true, out nrOfAlternatives, out curAlternative, out prices);
356
357       if (nrOfAlternatives == 0)
358       {
359         say (`Unfortunately I cannot find a journey at the time you provided|No journey found`);
360         say ("Please try again by specifying another time of travel");
361         set_unk (D_or_A, TravelDate, TravelMoment)
362         get_journey_plan = false;
363       }
364       else
365       {
366         optRmhNo = curAlternative; // optimal alternative
367
368         if (nrOfAlternatives == 1)
369           say ("I found @!<1> possible journey"); // @!<text>: text is emphasized
370         else
371           say ("I found a number of possible journeys");
372
373         get_journey_plan = true;
374       }
375     }
376
377
378     trigger ProvideJourneyDetails () precondition (get_journey_plan)
```

```
  activity travel_advice
  {
    string dTime;
    string aTime;
    string platformNo;

    ObtainJourneyDetails (curAlternative, out dTime, out aTime, out platformNo, out nrOfTrainChanges, out
interchangeStations, out journeyTime, out journeySchedHtml, out journeySchedText, out interchangeHtml, out
interchangeText);
    card ("Possible journey", null, journeySchedText, null, null); // BasicCard

    if (platformNo != null)
      say ("You will depart from {0} at {1} from platform {2}", Dstat [0], dTime, platformNo);
    else
      say ("You will depart from station {0} at {1}", Dstat [0], dTime);

    say ("You will arrive at station {0} at {1}", Astat [0], aTime);

    if (nrOfTrainChanges == 0)
      say "You do not have to change trains";
    else
      say ("You have to change trains {0} times", nrOfTrainChanges);

    if (journeyTime != null)
      say ("Your journey will take {0}", journeyTime);

    reported = true;
    get_journey_plan = false;
  }


  question AskRepeatJourneyDetails () precondition (reported)
  ask (nrOfTrainChanges == 0
       ? "Say @300 @!<earlier>, later, repeat or price"
       : "Zeg @300 @!<earlier>, later, repeat, transfers or price")
  suggest( nrOfTrainChanges == 0
           ? ['earlier', 'later', 'repeat', 'price', copyright]
           : ['earlier', 'later', 'repeat', 'transfers', 'price', copyright])
  { ( ( repeat | again )
      [get_journey_plan = true; reported = false ]
    | ( earlier
        [if (curAlternative == 0)
```

```
421             fail ("This was the earliest of the {0} journeys found", nrOfAlternatives);
422           say ("Now I will give an earlier journey");
423           curAlternative --
424         ]
425       | later
426         [if (curAlternative == nrOfAlternatives - 1)
427             fail ("This was the latest of the {0} journeys found", nrOfAlternatives);
428           say ("Now I will give a later journey");
429           curAlternative ++
430         ]
431       )
432       [ reported = false; get_journey_plan = true ]
433     | ( stop | halt | done | thanks | thank you )
434       [ user_done = true ]
435     )
436   }
437
438   // Patterns -- basically: reusable regular expressions that can have input and output parameters
439
440   pattern StationName (out string [] naam)
441   {
442     [ string [] s]
443     station? @s:<any>{1,4}
444     [ naam = CheckStation (s); if (naam == null) fail ($geenStation, s)]
445   }
446
447
448   // Date, Time etc. patterns: translated from Dutch more or less literally, not optimized for English
449
450   pattern Date (out DateTime date, out string dateString, out string dayName)
451   {
452     [int dayNo = -1; int offset; int weeksFromNow = 0]
453     ( ( today [offset = 0]
454       | tomorrow [offset = 1]
455       | the? day after tomorrow [offset = 2]
456       )
457       InNnnWeeks (out weeksFromNow)?
458       [date = CalcDate (offset, 0, weeksFromNow, <year>)]
459     | this?
460       NameOfDay (out dayNo)
461       [date = CalcDate (-1, dayNo, weeksFromNow, <year>)]
462     | NameOfDay (out dayNo)
```

```
463         next week // interpretation: first <...day> after this Sunday (Sunday can be today)
464         [date = CalcDate (-3, dayNo, weeksFromNow, <year>)]
465       | NameOfDay (out dayNo) // interpretation: first <...day> (can be today) plus <weeksFromNow> weeks
466         InNnnWeeks (out weeksFromNow)?
467         [date = CalcDate (-1, dayNo, weeksFromNow, <year>)]
468       | on?
469         MonthDate (out date)
470       )
471       [ dateString = DateToString (date);
472         dayName = DayName (date)
473       ]
474     }
475
476
477     pattern NameOfDay (out int dayNo)
478     {
479       ( sunday    [dayNo = 0]
480       | monday    [dayNo = 1]
481       | tuesday   [dayNo = 2]
482       | wednesdag [dayNo = 3]
483       | thursday  [dayNo = 4]
484       | friday    [dayNo = 5]
485       | saturday  [dayNo = 6]
486       )
487     }
488
489
490     pattern InNnnWeeks (out int n)
491     {
492       ( in a week   [n = 1]
493       | in @n:<int> weeks
494       )
495     }
496
497
498     pattern MonthDate (out DateTime date)
499     {
500       [int yearNo = <year>, monthNo, dayNo]
501       ( @dayNo:<int:1..31>
502         ( januari  [monthNo = 1]
503         | maart    [monthNo = 3]
504         | mei      [monthNo = 5]
```

```
      | juli      [monthNo = 7]
      | augustus  [monthNo = 8]
      | oktober   [monthNo = 10]
      | december  [monthNo = 12]
      )
    | @dayNo:<int:1..29>
      februari [monthNo = 2]
    | @dayNo:<int:1..30>
      ( april     [monthNo = 4]
      | juni      [monthNo = 6]
      | september [monthNo = 9]
      | november  [monthNo = 11]
      )
    )
    ( @yearNo:<int:1500..> )?
    [? monthNo == 2 && dayNo == 29 && IsLeapYear(yearNo) || monthNo != 2 || dayNo != 29 ?]
    [date = CalcDate (-2, dayNo, monthNo, yearNo)]
  }


  pattern Moment (bool isToday, out DateTime time, out string timeString)
  {
    [ hhh = <hour>;
      mmm = <minute>;
      int n;
      int dayPart = -1 // -1 => not (yet) specified
    ]
    ( ( at? ClockTime (out hhh, out mmm)
        PartOfDay (out dayPart)?
      | PartOfDay (out dayPart)
        at? ClockTime (out hhh, out mmm)
      )                                          // suppose now 15:00 and hhh = 9:00 => hhh = 21:00
      [ if (isToday && dayPart == -1 && hhh <= 12 && <hour> > hhh ||
            hhh <= 6 && dayPart == 1 ||                 // 2 o'clock in the afternoon => hhh = 14:00
            hhh >= 6 && hhh < 12 && dayPart == 2)       // 8 o'clock in the evening => hhh = 20:00
          hhh += 12;
      ]
    | [?isToday?] // guard, fails if date of travel is *not* today ('now' etc. pointless in that case)
      ( now
      | in
        MinutePeriod (out n)
        [hhh += n]
```

```
547         //| MinutePeriod (out n)
548         //  ago
549         //  [hhh -= n]
550         )
551       )
552       [ time = MakeDateTime (<today>, hhh, mmm);
553         timeString = TimeToString (time);
554       ]
555     }
556
557
558     pattern ClockTime (out int h, out int m)
559     {
560       [m = 0]
561       ( ( quarter (to [m = -15] | past [m = 15] )
562         | half [m = -30]
563         | @m:<int:1..29> ( minute | minutes )?
564           ( before [m = -m]
565           | ( past | after )
566           )
567         )
568         @h:<int:1..12>
569         [ if (m < 0) {h -= 1; m = 60+m}]
570       | @h:<int:0..24>  ( hours | ':' )
571         ( @m:<int:0..59> ( minute | minutes )? )?
572       )
573     }
574
575
576     pattern MinutePeriod (out int m)
577     {
578       [int n]
579       ( ( @n:<int> | one [n = 1] )
580         ( minute-s [m = n]   // minute-s will expand to ( minute | minutes ). Regexes are supported as well
581         | ( hour | hours ) [m = 60*n]
582         | quarter (of an hour)? [m=15*n]
583         )
584       | ( a half hour | half an hour ) [m = 30]
585       | ( 1 | one ) and a half hour [m = 90]
586       )
587     }
588
```

```
589
590     pattern PartOfDay (out int dayPart)
591     {
592       ( ( this | in the ) morning [dayPart = 0]
593       | ( this | in the ) afternoon [dayPart = 1]
594       | ( tonight | at night ) [dayPart = 2]
595       )
596     }
597
598
599     pattern YesNo (out bool b)
600     {
601       [string what]
602       ( yes <any>*
603         [b = true; say ("I heard yes")]
604       | no <any>*
605         [b = false; say ("I heard no")]
606       | @what:<any>+
607         [ b = false;
608           fail ("I heard {0}. Please say yes or no|Say yes or no", what)
609         ]
610       )
611     }
612
613     // Commands -- when an answer to a question is received, the dialog driver
614     //             first checks whether the answer is a command. If it is not, an
615     //             applicable follow-up question will be searched
616
617     command Directives ()
618     activity travel_advice, additional_options
619     {
620       ( ( i want )?
621         ( a ( other | different ) | to? change )
622         ( my | the )?
623         ( departure station
624           [?is_set (Dstat)?] // does only make sense if supplied earlier
625           [set_unk (Dstat);
626           ]
627         | ( arrival | destination ) station
628           [?is_set (Astat)?]
629           [set_unk (Astat);
630           ]
```

```
631       | date | ( date of ( departure | arrival ) | ( departure | arrival ) date )
632         [?is_set (TravelDate)?]
633         [set_unk (TravelDate); set_unk (TravelMomentDA);
634         ]
635       | time
636         [?is_set (TravelMoment)?]
637         [set_unk (TravelMoment); set_unk (TravelMomentDA);
638         ]
639       | ( time of departure | departure time )
640         [?all_set (TravelDate, TravelMoment)?] // does only make sense if supplied earlier
641         [D_or_A = true; set_unk (TravelDate); set_unk (TravelMoment); set_unk (TravelMomentDA);
642         ]
643       | ( time of arrival | arrival time )
644         [?all_set (TravelDate, TravelMoment)?]
645         [D_or_A = false; set_unk (TravelDate); set_unk (TravelMoment); set_unk (TravelMomentDA);
646         ]
647       )
648     | ( stop | done | halt )
649       [ user_done = true ]
650     | ( how long ( will | does ) the journey ( take | last ) // regexes are optimized, so don't
651       | how long will i be on the way                        // worry about common start sequences
652       )
653       [ if (nrOfAlternatives == 0)
654           say "No journey has been determined yet"
655         else
656           say ("Your journey will last {0}", journeyTime)
657       ]
658     | ( transfers
659       | ( how ( often | many times )? do i have to change trains )
660       )
661       [ if (nrOfAlternatives == 0)
662           say "No journey has been determined yet"
663         else  if (nrOfTrainChanges == 0)
664           say "You do not have to change trains";
665         else
666         {
667           say ("You will have to change trains {0} times", nrOfTrainChanges);
668           say ("You change trains at {0}", interchangeStations)
669           card (Format ("Interchange stations ({0})", nrOfTrainChanges), null, interchangeText, null, null);
670         }
671       ]
672     | [?prices != null && nrOfAlternatives > 0?] // guarded expression, must be true for this branch to succeed
```

```
673         ( ( price | prices )
674         | ( how much | what ) does ( it | ( the | this ) journey ) cost
675         )
676         [ if (FreeSubscription)
677             say ("You have a subscription and there will be no additional costs|")
678           else
679           {
680             say ("A 2nd class return ticket costs {0}| ", prices [1][0]);
681             say ("A 2nd class one-way ticket costs", prices [1][1]);
682             say ("A 1st class return ticket costs", prices [0][0]);
683             say ("A 1st class one-way ticket costs {0}", prices [0][1]);
684             table // show basic table with price info
685             (
686              "Prices for this journey",
687              "prices are without reduction",
688              [
689                { header: 'Class', align: 'LEADING' },
690                { header: 'Return', align: 'TRAILING' },
691                { header: 'One-way',  align: 'TRAILING' }
692              ],
693              [
694                { cells: ['2nd', prices [1][0], prices [1][1]] },
695                { cells: ['1st', prices [0][0], prices [0][1]] }
696              ],
697              null,
698              null
699             )
700           }
701         ]
702         )
703       | help
704         [ say("The following commands are available");
705           say("Previous or back @1 To repeat the previous question"); // @1 is one second break
706           say("Restart @1 To start all over");
707           say("Stop @1 To terminate the session");
708
709           if (nrOfAlternatives == 0)
710             say("@400 When a journey has been determined, the following commands are available")
711
712           say("Repeat @1 For the departure and arrival time");
713           say("Transfers @1 For information about changing trains");
714           say("Prices @1 For information about prices");
```

```
715         ]
716       )
717     }
718
719     command SystemCommands ()
720     {
721       ( start
722         ( stop | halt )
723         [say ("The session has been stopped"); halt]
724       | ( (repeat the?)? previous question? | back )
725         [redo]
726       | restart
727         [ say ("The session will restart");
728           restart
729         ]
730       | ( speakeasy | maker | copyright | @copyright )
731         [ say('This application has been developed by Speak Easy Software. @3|SpeakEasy Software')
732           image ("https://www.speakeasy.nl/images/speakeasy_logo.jpg", "SpeakEasy Software Amersfoort");
733         ]
734       | version
735         [say('The version number is <say-as interpret-as="ordinal">{0}</say-as>', <version>)]
736       )
737     }
738   }
739
```